Università di Genova | DIBRIS DIPARTIMENTO DI INFORMATICA, BIOINGEGNERIA, ROBOTICA E INGEGNERIA DEI SISTEMI

SurgiQ

UNIVERSITÀ DELLA CALABRIA

Istituti Clinici Scientifici Maugeri

A Two-Phase ASP Encoding for Solving Rehabilitation Scheduling

Matteo Cardellini[1,4], Paolo de Nardi[2], Carmine Dodaro[3], Giuseppe Galatà[1], Anna Giardini[2], Marco Maratea[4], and Ivan Porro[1]

[1] SurgiQ srl, Italy; {name.surname}@surgiq.com
[2] ICS Maugeri, Italy; {name.surname}@icsmaugeri.it
[3] DeMaCS, University of Calabria, Rende, Italy; dodaro@mat.unical.it
[4] DIBRIS, University of Genova, Genova, Italy; marco.maratea@unige.it

Context & Motivations

The rehabilitation scheduling process (RSP) consists of **planning patients' physiotherapy sessions inside a rehabilitation institute.**

A recent study found out that **almost one third of the people** in the world will need rehabilitation at some point during the course of their life [1].

In addition, **people affected by COVID-19 could manifest long-term consequences** (long-COVID), thus further increasing the demand for rehabilitation services globally.

[1] Cieza, A., Causey, K., Kamenov, K., Hanson, S.W., Chatterji, S., Vos, T.: Globalestimates of the need for rehabilitation based on the Global Burden of Disease study 2019. The Lancet 396 (10267), 2006–2017 (2020), Elsevier

Context & Motivations

The **RSP is subject to several constraints**, i.e., legal, medical and ethical, that need to be taken into consideration in order to find a viable schedule.

Until 2020 ICS Maugeri **managed the scheduling of physiotherapy sessions, in all its hospitals, by hand** and with little support to their decision.

Contribution

In this presentation, we exhibit a solution to the RSP based on **Answer Set Programming (ASP)** which is currently in production showing to be effecting in scheduling rehabilitation session in **hospitals of ICS Maugeri (Genova Nervi, Castel Goffredo).**

For the purpose of showing the scalability of the system, **benchmarks have been performed on larger synthetic instances**, whose parameters are inspired by the real data.

A Two-Phase Approach



The division in two phases is important to allow Maugeri's Coordinators to perform any desired manual change to the board, before planning the agenda

Board problem - First phase

Assign all patients to all the available operators

```
1 {assignment(OP, PAT) : operator(OP)} = 1 :- patient(PAT).
2 uniqueLocationLength(OP,PAT,DUR) :- assignment(OP,PAT), patient_session(PAT,_,LOC),
      patient_data(PAT,_,DUR), #count{ID:patient_session(ID,_,LOC), assignment(OP,ID)} < 2.
3 sameLocationLength(OP,PAT,DUR) :- assignment(OP,PAT), patient_session(PAT,DUR,LOC),
      #count{ID:patient_session(ID,_,LOC), assignment(OP,ID)} > 1.
4 :- operator_contract(OP,TIME,_), #sum{U,PAT:uniqueLocationLength(OP,PAT,U); S,
      PAT:sameLocationLength(OP,PAT,S)} > TIME.
5 :- operator_contract(OP,_,N), #count{PAT:assignment(OP,PAT)} > N.
6 :- operator_limit(OP,T,N), #count{PAT:assignment(OP,PAT), patient_data(PAT,T,_} > N.
7 :~ #sum{W, PAT:assignment(OP,PAT), patient_preference(PAT,OP,W)} = N. [N@3]
8 :~ #count{PAT: assignment(-1, PAT)} = N. [N@2]
9 :~ #sum{W, PAT:assignment(OP,PAT), history_preference(PAT,OP,W)} = N. [N@1]
```

Respect of operator's working time (Part-time, Full-time).

Respect of operator's qualifications.

Respect of patients' preferred operators.

Agenda problem - Second Phase



OBJECTIVE: Assign a starting time, duration and location to every (mandatory) rehabilitation session. Assign also as much optional sessions as possible.




```
{start(ID,PER,TS) : time(PER,OP,TS)} = 1 :- session(ID,_,OP), mandatory_session(ID).
{start(ID,PER,TS) : time(PER,OP,TS)} <= 1 :- session(ID,_,OP), optional_session(ID).
{length(ID,PER,NL) : time(PER,OP,L), NL=L-ST, TS+NL <= END, NL>= MIN, NL<= IDEAL} = 1 :-
    start(ID,PER,TS), period(PER,OP,ST,END), session(ID,_,OP),
    session_length(ID,MIN,IDEAL).
{session_location(ID,LOC): macro_location(MAC,LOC)} = 1 :- session_macro_location(ID,MAC).
{before(ID,NL): time(PER,OP,L), NL=L-ST, NL<=TS-ST} = 1 :- start(ID,PER,TS),
    period(PER,OP,ST,_), session(ID,_,OP).
{after(ID,NL): time(PER,OP,L), NL=L-ST, NL<=END-TS-LEN} = 1 :- start(ID,PER,TS),
    period(PER,OP,ST,END), length(ID,PER,LEN), session(ID,_,OP).
ext_start(ID,PER,TS-LB) :- start(ID,PER,TS), before(ID,LB).
ext_length(ID,PER,L+LA+LB) :- length(ID,PER,L), after(ID,LA), before(ID,LB).
```

Agenda problem - Second Phase

```
individual_session_location(ID,LOC,OP,MIN,IDEAL) :- session_type(ID,OP,individual),
    session_location(ID,LOC), session_length(ID,MIN,IDEAL).
session_time(ID,OP,PL,PER,TS..TS+L-1) :- session(ID,_,OP), session_location(ID,PL),
    ext_start(ID,PER,TS), ext_length(ID,PER,L).
:- start(ID,PER,TS), length(ID,PER,L), session_type(ID,OP,individual), start(ID2,PER,TS2),
    session_type(ID2,OP,individual), ID!=ID2, TS2>=TS, TS2<TS+L.
:- session(ID1,PAT,_), session(ID2,PAT,_), start(ID1,PER,_), start(ID2,PER,_), ID1!=ID2.
:- individual_session_location(ID1,LOC,OP,MIN1,OPT1), length(ID1,PER,L1),
    individual_session_location(ID2,LOC,OP,MIN2,OPT2), length(ID2,PER,L2), OPT1-L1 <=
    OPT2-MIN2, OPT2-L2 <= OPT1-MIN1 , |OPT1 -L1 - OPT2 + L2| > 1.
:- individual_session_location(ID1,LOC,OP,MIN1,OPT1), length(ID1,PER,L1),
    individual_session_location(ID2,LOC,OP,MIN2,OPT2), length(ID2,PER,L2), OPT1-L1 >
    OPT2-MIN2, L2 > MIN2.
:- individual_session_location(ID1,LOC,OP,MIN1,OPT1), length(ID1,PER,L1),
    individual_session_location(ID2,LOC,OP,MIN2,OPT2), length(ID2,PER,L2), OPT1-L1 <=
    OPT2-MIN2, OPT2-L2 <= OPT1-MIN1, OPT2 < OPT1, OPT1-L1 < OPT2-L2.
:- session_time(ID,OP,PL,PER,T), session_time(ID2,OP,PL2,PER,T), ID != ID2, PL != PL2.
:- patient(PAT,MIN), #sum{LEN, ID: session(ID,PAT,_), ext_length(ID,_,LEN)} < MIN.
:- location(LOC,LIM,PER,ST,END), LIM>0, time(PER,_,T), T>=ST, T<END, #count{ID:
    session_time(ID,_,LOC,PER,T)} > LIM.
:- forbidden(PAT,PER,ST,_), session(ID,PAT,_), ext_start(ID,PER,TS), ext_length(ID,PER,L),
    ST>=TS, ST<TS+L.
:- forbidden(PAT,PER,_,END), session(ID,PAT,_), ext_start(ID,PER,TS), ext_length(ID,PER,L),
    END>TS, END<=TS+L.
:- forbidden(PAT,PER,ST,END), session(ID,PAT,_), ext_start(ID,PER,TS),
    ext_length(ID,PER,L), ST<=TS,END>TS.
```

No overlap between two one-on-one sessions assigned to the same operator.

Avoid both sessions in the morning or in the afternoon

Optional supervised time must be assigned fairly

Compliance with the **minimum one-on-one length of the session.**

Observance of the **maximum capacity** of the locations.

Times in which the patient is unavailable must be respected

Agenda problem - Second Phase

```
:~ length(ID,_, L), session_length(ID,MIN,IDEAL), D=|L-IDEAL|. [D@6, ID]
:~ start(ID,PER,_), session_type(ID,_,individual), session_preference(ID,PER2,_,high),
    D=|PER-PER2|. [D@5, ID]
:~ start(ID,PER,TS), session_type(ID,_,individual), session_preference(ID,PER,TS2,high),
    D=|TS-TS2|. [D@4, ID]
:~ optional_session(ID), time(PER,_,TS), not start(ID,PER,TS). [1@3,ID]
:~ start(ID,PER,_), session_preference(ID,PER2,_,low), session_type(ID,_,individual),
    optional_session(ID), D=|PER-PER2|. [D@2, ID]
:~ start(ID,PER,TS), session_preference(ID,PER,TS2,low), session_type(ID,_,individual),
    optional_session(ID), D=|TS-TS2|. [D@1, ID]
```

The overall length of the session (one-on-one + supervised) should be **as close as possible to the ideal length** specified by the therapist

Respect, as much as possible, of the **preferred time and period** (specified or inferred).

The largest possible number of **optional sessions** should be included

Experimental Analysis

Table 1: Dimensions of the ICS Maugeri's institutes.

Institute	# Operators	# Patients	Density	# Floors	# Gyms
Genova Nervi	[9,18]	[37,67]	[2.4,5.2]	1	1
Castel Goffredo	[11,17]	[51,78]	[3.5, 6.4]	2	3

ICS Maugeri utilizes, in its daily activity of scheduling the rehabilitation session of its patients, a web-based software called QRehab, developed by SurgiQ, which is built on top of the encoding specified in this paper.

Experimental Analysis

Table 2: Results on ICS Maugeri institutes.

	Branch & Bound + RoM				Unsatisfiable Core			
	Genova Nervi		Castel Goffredo		Genova Nervi		Castel Goffredo	
	Board	Agenda	Board	Agenda	Board	Agenda	Board	Agenda
% Optimum	35%	0%	0%	0%	22%	45%	0%	0%
% Satisfiable	65%	100%	100%	67%	78%	55%	100%	70%
% Unknown	0%	0%	0%	33%	0%	0%	0%	30%
Avg Time for opt	1.1s	-	-	-	10s	0.01s	-	-
Avg Time Last SM	1.3s	30s	5.2s	30s	12.1s	21.3s	10.4s	30s

Branch & Bound + ROM = **clingo** encoding.lp --restart-on-model

Unsatisfiable Core = **clingo** encoding.lp --opt-strategy=usc,k,0,4 --opt-usc-shrink=bin

Cut Off: 30s

How does the problem scale ?



Scalability test - Board



These synthetic instances were produced **with features as close as possible to the ones of real hospitals**.
For example:

- the percentage of individual and supervised sessions,
- the medium length of operator's shifts,
- the occurrence of forbidden time slots for patients,
- the ideal length of sessions.

Scalability test - Board



Fig. 3: Results of CLINGO using the BB optimization algorithm (left) and the USC optimization algorithm (right) on synthetic benchmarks of the board.

Cut Off: 30s

Scalability test - Agenda



Scalability test - Agenda



Fig. 4: Results of the synthetic benchmarks of the agenda produced by CLINGO with the BB optimization algorithm and the option `--restart-on-model` enabled (left) and the USC optimization algorithm (right).

Cut Off: 30s

Which feature contributes more to the final result ?





This decision tree allows to predict the result of the **synthetic instances of the board** with few features and a **MAE of 6.3%**.

Validation of synthetic instances



Conclusions and Future Work

In this paper, we have presented a two-phase ASP encoding for solving rehabilitation scheduling.

Results **are positive for the institutes employed at the moment** and **give some positive indications on the upcoming ones.**

A possible topic for future research is to **improve the current encoding,** as well as **combining the strengths** of the optimizations algorithms employed.

Another interesting direction is to design also **rescheduling solutions,** to be applied in case of unavailability of operators and/or patients.

Thank you for your attention